
ABSTRACT  

Can you actually get something for nothing? With PROC SQL's subquery and remerging features, yes, you can.  
When working with categorical variables, often there is a need to add flag variables based on group descriptive 
statistics such as group counts, minimum and maximum values. Instead of first creating the group count, minimum or 
maximum values and then merging the summarized dataset to the original dataset with conditional statements 
creating a flag variable, why not take advantage of PROC SQL to complete three steps in one?  

With PROC SQL's subquery, CASE-WHEN clause and summary functions by the group variable, you can easily 
remerge the new flag variable back to the original dataset. 

INTRODUCTION 

While Proc SQL can be complex, it is also a powerful and concise programming language.  With a few years of 
experience and applying basic Proc SQL code on a daily basis, SAS programmers may soon realize that combining 
several benefits of Proc SQL in a single or related process makes a lot of sense to save program development time.  
One of Proc SQL‟s strengths is its ability to easily remerge with itself summary level descriptive statistics.  The 
technique presented in this paper is also applicable for adding any new variable to the dataset. 

PROC SQL ADVANTAGES 

By selecting all variables from the original data set and applying a LEFT JOIN, we are preserving the original dataset. 
In the subquery, an internal temporary dataset is created based on selecting both the linking variable and the new 
group descriptive statistics variable. The linking variable can uniquely identifies records or can be the grouping 
variable. A summary function is applied to the linking variable with the GROUP BY clause.  

For example, if there were 10 males and 9 females, then the SEXN value would be 10 for all males and 9 for all 
females.  The value of the linking variable also has a value for SEX, male or female, which is associated with the 
group count.  Conditions can be applied as needed, for example, to exclude missing values. Note that by including 
NAME in the subquery, one record per NAME value is returned.  The alternative is to not include the NAME variable 
and link by the grouping variable which returns only two records.  

The advantages of applying subqueries within remerge techniques is that the original dataset does not change in 
data content or record sort order. Without subqueries, however, if the GROUP BY clause is directly applied with the 
wildcard „*‟, which selects all variables, the original dataset may not be preserved. In general, this approach should be 
avoided since it may incorrectly cause values to be summarized or records to be rearranged. Directly applying the 
GROUP BY clause to get group descriptive statistics may be more appropriate for selected limited number of 
variables in the original dataset instead of applying it to all variables with the wildcard „*‟. 

SAS TECHNICAL TECHNIQUES OUTLINE 

Below is the outline to add group descriptive statistics as shown in Figure 1: 

1. Select all master variables to preserve original dataset values and sort order.
2. Include the new group descriptive statistics variables in the outer SELECT.
3. Apply LEFT JOIN to keep all master records.
4. Create internal dataset from the same master outer dataset using a subquery.
5. Select or create grouping variable.  Group variables are mostly categorical type variables which are

generally character values.  To create group descriptive statistics based on a continuous variable, first
create a format or IF-THEN logic to group the values into several categories.  For example, create a new
agegrp variable to group ages.

6. Apply on or more summary functions to get group descriptive statistics on one or more numeric variables.
7. Link internal dataset from subquery to master dataset by grouping or key variable.

proc sql; 
   create table class2 as 
   select   
    a.*,                                                           /* 1. Select  all master variables */ 
    b.sexn                                                      /* 2. Include the new group descriptive statistics variable */ 
   from sashelp.class as a 
   left join                                                      /* 3. Apply LEFT JOIN to keep all master records */ 
   (select                                                       /* 4. Create internal dataset using subquery */ 
    sex,                                                          /* 5. Select or create grouping variable */ 
    count(sex) as sexn                                  /* 6. Apply one or more summary functions */ 
   from sashelp.class  where sex > ' ' 
   group by sex 
   ) as b on  
  a.sex=b.sex;                                              /* 7. Link internal dataset by grouping variable */ 
quit; 

Figure 1. PROC SQL LINKED BY GROUP VARIABLE 
 
Because the subquery was linked by the grouping variable, the final dataset class2 is grouped by SEX. 

 

 
Output 1. PROC SQL LINKED BY GROUP VARIABLE 

 

SAS TECHNICAL TECHNIQUES STEPS 

 
1. Select all master variables, ex. A.* 
- Preserve original dataset values and sort order; 
 
2a, 2b. Apply one or more summary functions (COUNT, MIN, MAX, etc.) to get group descriptive stats on one or 
more numeric variables with the same grouping variable saved as one or more variables, ex. SEXN.  Practical 
application - DUPCNT 
 
3. Left join to keep all master records, ex. LEFT JOIN 
 
4. Subquery creates a temp dataset from the same outer master dataset, ex. () 
- SELECT limited variables 
- Any detail level condition such as grouping variable non-missing 
- Create new subquery for each new grouping variable 
- Apply HAVING for summary level condition, ex. HAVING VISIT=MAX(VISIT) 
- Does not have to be all inclusive of the linking variable 
- Could be same or different dataset 

- No semi-colon at the end 
 
5a, 5b, 7.  
Specify one or more grouping character variable (ex. SEX) or formatted numeric variable (ex. AGEGRP) 
- Create user defined format for AGEGRP is needed 
 
Two options for linking master and subquery datasets (1-to-M method may be more robust, both options have the 
same results)  
-  (One-to-Many) Join by summary variable (ex. SEX) which is many record per master dataset summary variable.  
Expect one record per unique summary variable. 
-  (One-to-One) Join by detail key variables (ex. NAME) which is one record per master dataset detail variable. 
Expect one record by unique detail variable. 
 
Figure 2 is an example of a one-to-one join by key variables to get SEX counts.  As you can see, the results are 
similar to figure 1 which was a one-to-many join by group variable.  While in this example, the key variable is only one 
variable, there may be other cases it may be more than one variable.  Linking by the grouping variable, however, 
offers you an advantage of linking by one variable. 
 

proc sql; 
   create table class3 as 
   select   
 1 - a.*,  
 2a - b.sexn 
   from sashelp.class as a 
 3 - left join 
 4  (select  
 5a - name,  
 2b - count(sex) as sexn 
   from sashelp.class  where sex > ' ' 
 7 - group by sex 
   ) as b on  
 5b - a.name=b.name; 
quit; 

Figure 2. PROC SQL LINKED BY KEY VARIABLE 

 
Because the subquery was linked by the key variable NAME, the final dataset class3 is sorted by NAME.  Notice, 
however, that the group descriptive statistics are still the same as when the subquery was linked by SEX. 

 

 
Output 2. PROC SQL LINKED BY KEY VARIABLE (Same results as Output 1. PROC SQL LINKED BY GROUP 
VARIABLE) 

Figure 3 is an example of multiple group descriptive statistics joined by grouping variable.  SEX counts, minimum and 
maximum WEIGHT by SEX are added to the dataset.  Note that descriptive statistics can be based on variables other 
than the grouping variable.  Note also that with this simple technique, descriptive statistics by other grouping 
variables can easily be added by repeating lines 3 to 8 and adding the new variable in line 2.  

 
proc sql; 
   create table class4 as 
   select   
 1 - a.*,  
 2 - b.sexn2, b.minwgt, b.maxwgt 
   from sashelp.class as a 
 3 - left join 
 4  (select  
 5 - sex,  
 6 - count(sex) as sexn2, min(weight) as minwgt, max(weight) as maxwgt 
   from sashelp.class  where sex > ' ' 
 7 - group by sex 
   ) as b on  
 8 - a.sex=b.sex; 
quit; 

Figure 3. PROC SQL WITH MULTIPLE GROUP DESCRIPTIVE STATISICS VARIABLES 
 
For each SEX group, the minimum and maximum weight by SEX is added as MINWGT and MAXWGT variables.  As 
expected, the values are the same within each SEX group. 

 

 
Output 3. PROC SQL WITH MULTIPLE GROUP DESCRIPTIVE STATISICS VARIABLES 

 
Figure 4 is an example of numeric group descriptive statistics.  The data step before Proc SQL creates the numeric 
grouping variable, AGEGRP based on AGE values.  Proc SQL adds the counts for AGEGRP. 

 
data class1; 
 set sashelp.class; 
 if age < 15 then agegrp = „LT 15‟; 
 else if age >= 15 then agegrp=‟GE 15‟; 
run; 
 
proc sql; 
   create table class5 as 
   select   
 1 - a.*,  
 2 - b.agen  
   from class1 as a 
 3 - left join 
 4  (select  
 5 - agegrp,  

 6 - count(age) as agen 
   from class1  where age > . 
 7 - group by agegrp 
   ) as b on  
 8 - a.agegrp=b.agegrp; 
quit; 

Figure 4. PROC SQL WITH NUMERIC GROUP DESCRIPTIVE STATISICS VARIABLE 

 
For continuous variables such as AGE, once a user defined format is applied to group into categories of 2 for 
example, the group counts AGEN can be added.  In the example below, we see that there are 5 people with ages 
greater than or equal to 15 and 14 people with ages less than 15. 

 

 
Output 4. PROC SQL WITH NUMERIC GROUP DESCRIPTIVE STATISICS VARIABLE 

 
Figure 5 is an example of group descriptive statistics based on another variable.  Since grouping and descriptive 
statistics are related yet independent of each other.  Proc SQL enables you to group by one variable such as 
PRODUCT and calculate descriptive statistics based on another variable such as SALES.  In this typical business 
example, we want SALES descriptive statistics, count, minimum and maximum by each PRODUCT.  Notice that the 
linking variable is the PRODUCT grouping variable.  

 
proc sql; 
   create table class6 as 
   select   
 1 - a.*,  
 2 - b.salesn, b.salesmn, b.salesmx 
   from sashelp.shoes as a 
 3 - left join 
 4  (select  
 5 - product,  
 6 - count(sales) as salesn, min(sales) as salesmn, max(sales) as salesmx 
   from sashelp.shoes  where sales > . 
 7 - group by product 
   ) as b on  
 8 - a.product=b.product; 
quit; 

Figure 5. PROC SQL WITH GROUP DESCRIPTIVE STATISICS BASED ON ANOTHER VARIABLE 
 
Below, you can see the number of SALES records, SALESN, minimum SALES, SALESMN and maximum SALES, 
SALESMX for each PRODUCT.  As expected, the values are the same, for example, within the Boot PRODUCT.  
There is a change in SALESN, SALESMN and SALESMX values for Men‟s Casual PRODUCT.  While creating these 
variables, formats can be applied to add „$‟ before the SALESMN and SALESMX values.  

Output 5. PROC SQL WITH GROUP DESCRIPTIVE STATISICS BASED ON ANOTHER VARIABLE 

 
 
For a visual representation of Proc SQL‟s suq-queries, see the mind map below.  This mind map is part of a larger 
collection of interrelated mind maps.  For this and other free SAS mind maps, visit www.SASSavvy.com.  

 

SUMMARY 
 
As you can see, with a few simple techniques combined together, a Proc SQL template can be used on a regular 
basis to always easily add group descriptive statistics to your dataset.  This technique can be duplicated for multiple 
group descriptive statistics. Without the need for a second DATA step or creating a second dataset, you can now 
quickly add group descriptive statistics and move forward to the next by-group processing step. 
 
I hope you now have some insights into how and why Proc SQL is one of the best ways to accomplish this task.  
Based on when I first developed this technique, I realized that I could expand on it to build a more ultimate Proc SQL 
example with the HAVING, GROUP BY, and the WHERE clauses along with the subquery as shown in figure 6.  Note 
that caution must be applied to prevent the assumption that the HAVING clause impacts the GROUP BY clause when 
calculating mean values.  The HAVING clause is applied to subset the dataset only AFTER calculating the mean. 

 
proc sql; 
create table base2 as                             /* Add suffix number to base dataset */ 
select unique a.*, b.aval, b.dtype           /* Keep all A.* & new B.vars */ 
from base (drop=aval) as a                    /* Drop AVAL since replaced */ 

http://www.sassavvy.com/

right join                                                 /* Save AVERAGE records only */ 
(                                                              /* Subquery start */ 
select unique usubjid, paramn, ady,      /* Keep linking variables */  
 mean(aval) as aval,                               /* Summarize to get one rec/key */ 
 „AVERAGE‟ as dtype                            /* Flag data type method */ 
 from base                                              /* FROM same dataset as outer */ 
 where ady <= 1                                     /* Subset only baseline records */ 
 group by usubjid, paramn                     /* GROUP BY key variables for MEAN() */ 
 having max(ady) = ady                         /* Of the records, keep only max ADY */ 
)                                                              /* Subquery stop */ 
as b on a.usubjid=b.usubjid and a.paramn=b.paramn and a.ady=b.ady  
                                                               /* Linking variables with outer dataset */ 
; 
quit; 

Figure 6. EXAMPLE OF ULTIMATE PROC SQL CODE 

 
Note that with repeated subqueries to update an existing variable already in the master dataset, it is recommended to 
save results to a new variable such as B.XXX instead of using COALESC() function to keep the non-missing value.  
After the PROC SQL step, use DATA step to replace the original variable if the B.XXX variable is non-missing.  This 
technique takes advantage of PROC SQL to do the major task and DATA step to adjust the variable.  

 
proc sql; 
create table base2 as                                /* Add suffix number to base dataset */ 
select unique a.*, b.b_aval, b.b_dtype      /* Keep all A.* & new B.vars */ 
from base as a                                          /* Drop AVAL since replaced */ 
left join                                                       /* Save AVERAGE records only */ 
(                                                                /* Subquery start */ 
select unique usubjid, paramn, ady,        /* Keep linking variables */  
 mean(aval) as b_aval,                             /* Summarize to get one rec/key */ 
 „AVERAGE‟ as b_dtype                           /* Flag data type method */ 
 from base                                                 /* FROM same dataset as outer */ 
 where ady <= 1                                        /* Subset only baseline records */ 
 group by usubjid, paramn                        /* GROUP BY key variables for MEAN() */ 
 having max(ady) = ady                            /* Of the records, keep only max ADY */ 
)                                                                 /* Subquery stop */ 
as b on a.usubjid=b.usubjid and a.paramn=b.paramn and a.ady=b.ady  
                                                                  /* Linking variables with outer dataset */ 
; 
quit; 
 
data base2; 
 set base2; 
 
 if b_aval > . then aval=b_aval; 
 if b_dtype >  „ „ then dtype=b_dtype; 
run; 

Figure 7. EXAMPLE OF ULTIMATE PROC SQL CODE SAVING TO NEW VARIABLES  

